

Prolog's Cut Operator

Dr. Mattox Beckman

UNIVERSITY OF ILLINOIS AT URBANA-CHAMPAIGN
DEPARTMENT OF COMPUTER SCIENCE

Objectives

You should be able to ...

Prolog's greatest strength is its ability to backtrack to find alternative solutions. If not controlled, it can also be its greatest weakness. In this lecture, we will go over the cut operator, which gives a solution to this problem.

- ▶ Show how the cut operator stops backtracking.
- ▶ Use the cut operator to assert failure.
- ▶ Give an example of the limit of the cut operator.



Backtracking

```

1 color(red).
2 color(blue).
3 car(honda).
4 car(ford).
5 car(toyota).
6
7 ?- color(A), car(B).

```

```

1 A = red
2 B = honda ;
3 A = red
4 B = ford ;
5 A = red
6 B = toyota ;
7 A = blue
8 B = honda ;
9 A = blue
10 B = ford ;
11 A = blue
12 B = toyota ;

```

Come and see the backtracking inherent in the system!



The Cut Operator

- ▶ The cut operator (!) stops backtracking.
- ▶ It is considered a goal that always succeeds.

```

1 ?- color(A), !, car(B).
2
3 A = red
4 B = honda ;
5 A = red
6 B = ford ;
7 A = red
8 B = toyota ;
9 No

```

Commitment

Once a cut is activated, the clause we are trying to satisfy is committed to that choice.

```

1 color(red).
2 color(green) :- !.
3 color(blue).
4
5 ?- color(X).
6 X = red ;
7 X = green ;
8 No

```

Once X was set to green, the cut operator forces us to stay with green or else color should fail completely. **Question:** Can color(blue) ever be matched?

Factorial Revisited

```

1 fact(0,1).
2 fact(N,X) :- M is N-1, fact(M,Y),
3             X is Y * N.
4
5 ?- fact(5,N).
6
7 N = 120 ;
8 ERROR: Out of local stack

```

► What happened here?



Two Fixes

You can add a constraint to the second clause....

```

1 fact(0,1).
2 fact(N,X) :- N > 0, M is N-1, fact(M,Y),
3             X is Y * N.

```

Or you can add a cut to the first clause.

```

1 fact(0,1) :- !.
2 fact(N,X) :- M is N-1, fact(M,Y),
3             X is Y * N.

```

Now it will work:

```

1 ?- fact(5,N).
2 N = 120 ;
3 No

```



Efficiency

Suppose you run the campus observatory. You want to allow certain people to use the telescope. They have to be a student, a faculty member, or a member of the astronomy club. And they also need to have been trained on the telescope.

- students anna, beth, cindy, david
- faculty ernest, frank, gloria
- astronomy club anna, frank, harry
- trained anna, harry



Can frank Use the Telescope?

```

1 telescope(X) :- (student(X); faculty(X); club(X)),
2                 trained(X).

```

- ▶ frank is a faculty, and also a member of the club.
- ▶ But, frank doesn't have any training.

What will telescope(frank). do?

Who Can Use the Telescope?

```

1 telescope(X) :- (student(X); faculty(X); club(X)),
2                 trained(X).
3
4 ?- telescope(X).
5 X = anna ;
6 X = anna ;
7 X = harry ;

```

Since anna is a student *and* a member of the club, she gets listed twice.



Now We Cut ...

```

1 telescope(X) :- (student(X); faculty(X); club(X)),
2                 !,
3                 trained(X).
4
5 ?- telescope(X).
6 X = anna ;
7 No

```

- ▶ Oops. Now we've dissed harry.
- ▶ But at least we don't spend a lot of time when we ask if frank can use the telescope....
- ▶ Moral: cut will limit your choices to only one answer.

Inducing Failure

- ▶ We also have a predicate called fail, which, well, always fails.
- ▶ Suppose anna has her telescope privileges revoked....

```

1 telescope(anna) :- fail.
2 telescope(X) :- (student(X); faculty(X); club(X)),
3                 trained(X).
4
5 ?- telescope(anna).
6 Yes.

```

This is less than what we hoped for.



Inducing Failure, II

```
1 telescope(anna) :- !, fail.
2 telescope(X) :- (student(X); faculty(X); club(X)),
3                 trained(X).
4
5 ?- telescope(anna).
6 No
7 ?- telescope(harry).
8 Yes
9 ?- telescope(X).
10 No
```

But cut and fail will work.

Conclusions

- ▶ Cut can stop searches that you already know will be useless.
- ▶ Cut can make queries more efficient.
- ▶ But, cut can make queries do strange things. Use with care.

Aside: you can define `not` (actually, it's built in) this way:

```
1 not(X) :- call(X), !, fail.
2 not(X).
```

This predicate can fix the telescope problems.